

Puissances dans un amplificateur

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDF](#)

Cet article montre le fonctionnement d'un amplificateur de puissance de structure basique : émetteur commun et push pull de sortie. On s'intéressera plus particulièrement à la répartition de la puissance selon le niveau de signal.

1) L'amplificateur de puissance

Le schéma est classique, supposé connu, et, pour nos simulations, réalisé avec des composants de la librairie eval.lib, ce qui permet une étude générale, avec des valeurs numériques cohérentes.


1.a Présentation brève.

Une amplification en tension est réalisée par Q₄. En supposant ce transistor polarisé en mode normal (par R₇, R₈), et si on néglige le courant prélevé par les bases de XQ₁, XQ₂, l'amplification en tension est $-R_3/R_4$, soit ici - 22.

Le montage à base de Q₃ est un translateur de tension, qui polarise XQ₁, XQ₂. Son impédance dynamique est très faible.

L'étage de sortie est un push pull, réalisé ici par des Darlington complémentaires, ce qui permet d'avoir des courants de base - relativement - faibles.

R₁ et R₂ évitent l'emballement thermique. Les valeurs, très faibles, sont légèrement différentes pour contre balancer la non symétrie NPN PNP des Darlington.



netliste en annexe 1 `ampli.cir`

R ₁ = 0,1 Ω	R ₂ = 0,15 Ω	R ₃ = 2,2 kΩ
R ₄ = 100 Ω	R ₅ = 2,2 kΩ	R ₆ = 2,2 kΩ
R ₇ = 47 kΩ	R ₈ = 1,5 kΩ	C ₁ = 10 μF
Q ₃ = BD239	Q ₄ = 2N2222	
VPLUS = 50 V	VMOINS = -50 V	R _L = 8 Ω

1.b Simulation, à T = 27 °C

Dans un premier temps, on simule tout le montage à la température par défaut (300 K, 27 °C).

L'analyse .OP donne :

Offset, V(S) = -0,1078 V. Par loi d'Ohm, on trouve un courant de 13,5 mA dans R_L.

Les courants de polarisation des Darlington peuvent être déterminés par $\frac{V(E1) - V(S)}{R_1} =$

$$\frac{-0,1076 - (-0,1078)}{0,1} = 2 \text{ mA}, \text{ et } \frac{V(S) - V(E2)}{R_2} = \frac{-0,1078 - (-0,1102)}{0,15} = 16 \text{ mA}.$$

On est, à cette température, en limite de conduction.

On peut relever également $I_C(Q_4) = 22,5 \text{ mA}$ et $I_C(Q_3) = 21,8 \text{ mA}$, $V_{CEQ3} = V(B1) - V(B2) = 1,78 \text{ V}$.

En signal sinusoïdal, les transistors de puissance vont conduire à tour de rôle, chacun délivrant une demi-alternance. C'est pourquoi, une **analyse AC** n'est pas significative, car elle est réalisée par PSpice après linéarisation autour du point de repos .OP.

Pour prédéterminer la réponse harmonique :

- on peut estimer la coupure basse par l'équation : $\frac{1}{2\pi R_g C_1} = \frac{1}{2\pi \cdot 1,5k \cdot 10\mu} = 10,6 \text{ Hz}$,


si on néglige l'intervention des autres résistances environnantes,

- on peut estimer l'amplification en tension légèrement inférieure à -22 , pour tenir compte d'une atténuation par le push pull (collecteur commun) et les résistances R_1, R_2 . Soit un gain de 26 dB environ.


Analyse .TRAN, avec $V(\text{in}) = 1,5 \text{ V}$, 1 kHz.

Remarque : ce choix d'une amplitude d'entrée de 1,5 V est justifié dans le paragraphe suivant.

Pour mieux visualiser la polarisation, le signal est volontairement appliqué à $t = 1 \text{ ms}$.



Tension d'entrée, tension de sortie



courants dans les Darlington

Interprétation

Dès l'origine, on visualise la tension d'offset : - 107,8 mV.
C'est la valeur déterminée par l'analyse .OP.

X Values	500.000u
V(S)	-107.808m

A $t = 0$, les courants sont de 2 mA pour XQ₁ et 15,5 mA pour XQ₂, valeurs déjà déterminées dans l'analyse .OP.

X Values	500.000u
I(XQ1:1)	2.0607m
- I(XQ2:1)	15.482m

Remarque : soumis à environ 50 V, cela représente une puissance de 0,1 W et 0,775 W de dissipé par la polarisation.

Après un temps de stabilisation (vers 1 s), le signal de sortie évolue entre 29,613 V et - 30,173 V, soit 59,786 V crête à crête. Hors offset, si on suppose l'onde sinusoïdale, l'amplitude est de 29,893 V, la valeur efficace est 21,137 V. L'amplitude du signal d'entrée étant de 1,5 V, l'amplification en tension est de $29,893/1,5 = 19,93$, en opposition de phase. On est proche de la valeur prédéterminée.

Les courants dans les Darlington sont presque équilibrés. L'amplitude est 3,7 A pour XQ₁ et 3,77 A pour XQ₂.

Il existe un courant continu qui traverse les Darlington, en absence de signal. C'est le point délicat de cette structure en classe AB : il faut les transistors de puissance en limite de conduction, pour éviter le défaut de raccordement. C'est l'étage à base de Q₃ (translateur de niveau) qui règle ce courant, et c'est R₁ et R₂ qui règlent l'équilibre pour les demi-alternances.

Bilan des puissances

Après simulation, on peut afficher les puissances temporelles (instantanées), et, par la fonction AVG, leur valeur moyenne pour obtenir les valeurs efficaces.

X Values	1.0000
AVG(W(RL))	55.977
AVG(W(XQ1))	31.247
AVG(W(XQ2))	30.395
-(AVG(W(VDC1puiss))+ AVG(W(VDC2puiss)))	122.039
AVG(W(R1))	352.071m
AVG(W(R2))	1.6355

puissances efficaces dans l'étage de sortie

Remarque : il a été choisi un run de 1 s pour que les courbes des valeurs moyennes se stabilisent.

Pour cette **amplitude d'entrée de 1,5 V**, on a ainsi 56 W dans la résistance de 8 Ω.

C'est quasiment la valeur déterminée par la loi de Joule : $U_{\text{eff}}^2/R_L = (21,137)^2/8 = 55,85 \text{ W}$.

L'alimentation double fournit 122 W.

Les résistances R₁, R₂, bien que faibles, dissipent de la puissance : 0,352 W et 1,64 W respectivement.


Faisons le bilan : $P_{Q1} + P_{Q2} + P_{RL} + P_{R1} + P_{R2} : 31,25 + 30,4 + 56 + 0,35 + 1,64 \approx 119,6 \text{ W}$.

On retrouve ce que fournit l'alimentation, à 2 W près, dissipé dans le reste du montage.

D'ailleurs, il est fréquent d'associer la puissance délivrée par les alimentations à celle de l'étage de puissance.

Le rendement de l'étage de puissance est $56 / 122 \approx 46 \%$.

1.c Estimation des températures




TO3



TO220

Sur le dissipateur, le transistor BD239, en boîtier TO220, est placé entre les 2 Darlingtons de puissance de boîtier TO3.




les 2 TO3 et le TO220 montés sur un dissipateur (non à l'échelle)

Ce couplage thermique permet une protection : le BD 239 réduit la polarisation en cas de surchauffe des Darlingtons.

Avec une amplitude du signal d'entrée de 1,5 V, nous avons déterminé que le push pull dissipe :

$$P_{Q1} = 31,25 \text{ W et } P_{Q2} = 30,4 \text{ W.}$$

Entre la jonction et l'air ambiant de chacun, on peut définir 3 résistances thermiques : jonction - case (boîtier), case - surface, et surface - air ambiant.



2 Darlington côte à côte

- On admet que T_A , température de l'air ambiant au voisinage du dissipateur commun, est la même pour les 2 transistors du push pull.
- Par un placement symétrique des 2 transistors, on a les R_{THSA} identiques. Les dimensions et la forme du dissipateur donnent $R_{THSA} = 1 \text{ °/W}$. Valeur type d'un Fischer SK197, de longueur 200 mm.
- On peut estimer un contact thermique boîtier – surface du dissipateur R_{THCS} identique pour les 2 transistors, de valeur $0,8 \text{ °/W}$, (fonction du serrage, de l'état de surface).
- Chaque Darlington dispose de $R_{THJC} = 1,17 \text{ °/W}$ (données constructeur).
- Simplifions l'étude en supposant $P_{Q1} = P_{Q2}$. On a $P_{Q1} + P_{Q2} \approx 62 \text{ W}$.

Dans ces conditions de parfaite symétrie, les 2 chemins thermiques ont les mêmes T_A, T_S, T_C, T_J , et donc les résistances thermiques sont en parallèle.


On déduit alors un schéma thermique unique qui modélise le push pull.

On a : $P = 62 \text{ W}$, R_{THJC} arrondie à $0,6 \text{ °/W}$, $R_{THCS} = 0,4 \text{ °/W}$ et $R_{THSA} = 0,5 \text{ °/W}$.

On pose $R_{TH} = R_{THJC} + R_{THCS} + R_{THSA} = 1,5 \text{ °/W}$.

La loi d'Ohm thermique donne $\Delta T = R_{TH} P = 1,5 \text{ °/W} \times 62 \text{ W} = 93 \text{ °C}$.

Si on suppose une température ambiante, au voisinage du radiateur, de 50 °C , on aboutit à une température de jonction des Darlingtons de 143 °C .



Modèle équivalent du push pull.

Remarque : c'est un cas simplifié, mais cohérent, pour faciliter la compréhension de ce document. De plus, on n'a pas tenu compte des capacités thermiques, cette étude n'est valable qu'en statique.


Du schéma précédent, on obtient aisément la température du boîtier, T_C , par la somme des températures :

$$T_A + (R_{THSA} + R_{THCS}) \times P = 50 + 0,9 \times 62 = 105,8 \text{ °C}$$

ou par :

$$T_J - R_{THJC} \times P = 143 - 0,6 \times 62 = 105,8 \text{ °C.}$$

Remarque : la data sheet du 2N3459 (ou de son complémentaire) donne une puissance max de 80 W à cette température, ce qui garantit une marge de sécurité car il dissipe 31 W . Le dissipateur est largement dimensionné.



La puissance dissipée par le BD239 étant très faible, on peut négliger l'élévation de sa température due à son auto échauffement. Dans ce cas, sa température de jonction est égale à celle de son boîtier, donc T_C .

En conclusion, si le push pull dissipe la puissance de 62 W (31 W + 31 W), la température de jonction de chaque Darlington sera 143 °C, à $T_A = 50$ °C au voisinage du dissipateur, et la jonction du translateur de tension Q_3 sera de 106 °C.

1.d Simulation en température

Nous allons reprendre et approfondir les simulations avec les conditions suivantes : XQ_1, XQ_2 à 143 °C, BD239 à 106 °C et le reste du circuit à 27 °C. (Avouons qu'une telle précision sur les températures n'est pas nécessaire...)

PSpice permet d'imposer une température sur des semiconducteurs : il suffit d'ajouter, dans la liste des paramètres définissant le modèle, le paramètre **T_ABS**.

Pour éviter de placer ce paramètre dans les sous-circuits formant les Darlington, on a choisi une autre stratégie : on indique par la directive **.temp=143**, que le circuit simulé est à cette température, et on ajoute le paramètre **T_ABS=106** pour Q_3 , translateur de niveau, et **T_ABS=27** pour Q_4 (2N2222).


Les composants passifs, n'ayant pas de coefficient de température associé, ne sont pas impactés et sont simulés à 27 °C, valeur par défaut.

Dans ces conditions **l'analyse .OP** donne : $V(S) = -0,1411$ V, d'où 18 mA dans R_L .


$$I_C(Q_1) = \frac{V(E1) - V(S)}{R_1} = \frac{-0,1402 - (-0,1411)}{0,1} = 9 \text{ mA}, \quad I_C(Q_2) = \frac{V(S) - V(E2)}{R_2} = \frac{-0,1411 - (-0,145)}{0,15} = 26 \text{ mA}.$$

On peut relever également $I_C(Q_4) = 22,5$ mA (inchangé) et $I_C(Q_3) = 22,1$ mA.

L'analyse .TRAN donne :



Tension d'entrée, tension de sortie



courants dans les Darlington

Interprétation

A $t = 0,5$ ms $I_C(Q_1) = 8,61$ mA et $I_C(Q_2) = 26,22$ mA : on confirme **.OP**. On est en limite de conduction.

En régime établi, la tension $V(S)$ va de -30,05 V à 30 V, l'amplification en tension est $60,05/3 \approx 20$, en opposition de phase.

Les amplitudes des courants sont 3,75 A pour XQ_1 et 3,756 A pour XQ_2 . Le signal de sortie est quasiment symétrique.

Bilan des puissances

A 143 °C pour le push pull et 106 °C pour le DB239, les puissances sont peu changées par rapport à 27 °C.

Rappel :

la sinusoïde d'entrée a pour amplitude 1,5 V.

X Values	1.0000
AVG(W(RL))	56.839
AVG(W(XQ1))	31.303
AVG(W(XQ2))	30.958
-(AVG(W(VDC1puiss))+AVG(W(VDC2puiss)))	122.424
AVG(W(R1))	355.949m
AVG(W(R2))	531.813m

puissances efficaces

2) Répartition des puissances en fonction du niveau de sortie

Dans ce genre de configuration push pull en classe AB, contrairement à une première idée intuitive, la puissance dissipée par les transistors n'est pas maximale quand le niveau de sortie est maximum.

La théorie est très bien expliquée par M. Corveon, sous : <https://www.chireux.fr/mp/cours/electronique/Chap2.pdf>

Le signal de sortie s'écrit $U_{max} \sin \omega t$.

→ La puissance dans la charge est, par loi de Joule, $P_{RL} = \frac{U_{eff}^2}{R_L} = \frac{U_{max}^2}{2R_L}$.


→ La puissance dissipée dans les 2 transistors s'écrit : $P_Q = \frac{2V_{CC} U_{max}}{\pi R_L} - \frac{U_{max}^2}{2R_L}$.

Cette fonction passe par un maximum quand l'amplitude $U_{max} = \frac{2V_{CC}}{\pi}$. Dans ce cas, cette puissance vaut $\frac{2V_{CC}^2}{\pi^2 R_L}$.

→ La puissance fournie par les alimentations est $P_{RL} + P_Q = \frac{2V_{CC} U_{max}}{\pi R_L}$.

Cette fonction est maximale pour $U_{max} = V_{CC}$. Dans ce cas, elle vaut : $\frac{2V_{CC}^2}{\pi R_L}$.

Ce qui se résume à :



Interprétation

En abscisse, le niveau de sortie est en coordonnées réduites :

- pour 0, l'amplitude U_{max} vaut 0,
- pour 1, U_{max} vaut la tension d'alimentation, soit 50 V dans notre simulation.

En ordonnée, les puissances sont comparées par rapport à la puissance totale avec le niveau maximal.

On repère nettement le maximum de puissance dissipée par les transistors. Dans notre cas, où $V_{CC} = 50$ V, cela correspond à $V_{max} = 32$ V. Ramenée à l'entrée avec une amplification en tension de 22 théorique, cela correspond à une amplitude du signal d'attaque de 1,5 V environ.

Remarque :

il s'agit d'une étude théorique avec des transistors parfaits polarisés en classe AB. Et ce, en linéaire, donc pas de saturation ni distorsion. On suppose ainsi que l'amplitude de sortie peut atteindre VCC.

Nous allons vérifier cela par des simulations


Mode opératoire : par la directive .STEP, on lance des multiples runs en faisant croître l'amplitude de la sinusoïde d'entrée, à partir de 0 et par pas de 0,2 V, ce qui fait varier l'amplitude de la sinusoïde de sortie. On surveille la forme d'onde de cette dernière, on vérifie que l'amplification en tension est constante à 20, ce qui confirme la linéarité, et on arrête les essais quand une saturation apparaît en sortie.

Nous reprenons la fonction AVG et une durée de run de 1 s.

Les résultats sont ensuite récapitulés sur ce graphe :

En abscisse, la variable est l'amplitude de la sinusoïde d'entrée (la directive .STEP). Comme l'amplification en tension est constante, c'est donc l'image de l'amplitude de sortie.

Il est donc facile de comparer avec le graphe donné par M. Correvon :



Prenons le point à 1,8 V, soit 36 V en sortie.

→ La puissance dissipée dans R_L répond bien à la loi de Joule : $\frac{U_{\max}^2}{2R_L} = 81 \text{ W}$, ce que confirme la simulation (80,54 W).

→ La puissance totale répond à l'équation linéaire $\frac{2V_{CC}U_{\max}}{\pi R_L} = \frac{2 \times 50 \times 36}{\pi \times 8} = 143 \text{ W}$. La simulation donne 145,4 W pour les 2 alimentations. On observe un léger décalage à l'origine qui est la consommation du montage des différentes composantes continues.

→ La puissance dissipée par les transistors passe bien par un maximum, placé ici au voisinage de 1,5 V d'amplitude d'entrée. La théorie donne $\frac{2V_{CC}^2}{\pi^2 R_L} = \frac{2 \times 50^2}{\pi^2 \times 8} = 63,3 \text{ W}$. La simulation donne (paragraphe 1.d)

31,3 W + 30,96 = 62,26 W. Rappelons que nous avons ici des Darlington équipés de résistances internes.

L'avant dernier point (à amplitude d'entrée à 2,2 V) donne une amplification de -19 : on commence à quitter le domaine de linéarité.

Le dernier point (à amplitude d'entrée à 2,4 V) est placé pour avoir une meilleure étendue des traces, mais est légèrement faux : une saturation en V(S) commence à se voir : le signal est écrêté à - 43,109 V et à 43,721 V. La forme d'onde n'étant plus sinusoïdale, les équations de puissance précédentes ne peuvent plus s'appliquer.

CONCLUSION

Sur un schéma standard d'ampli 120 W sur 8 Ω , ce document a mis l'accent sur l'aspect thermique : dissipation de puissance, échauffement et dissipateur.

Mais l'aspect le plus marquant est de montrer, par la simulation, la répartition des puissances selon le niveau sonore : on a pu vérifier la fonction parabolique de la puissance dissipée par le push pull en fonction de l'amplitude du signal d'entrée.

Annexe : fichier ampli.cir

```
Amplificateur basique
* fichier ampli.cir
.lib eval.lib

.MODEL BD239_27      NPN(Is=2.447p Xti=3 Eg=1.11 Vaf=100 Bf=208.2 Ise=70.69p
+ Ne=1.565 Ikf=.9743 Nk=.6134 Xtb=1.5 Br=12.59 Isc=11.68n
+ Nc=1.835 Ikr=3.86 Rc=.4685 Cjc=142p Mjc=.4353 Vjc=.75 Fc=.5
+ Cje=188.5p Mje=.4878 Vje=.75 Tr=194.2n Tf=19.85n Itf=164.1
+ Xtf=5.945 Vtf=10 Rb=.1 T_ABS=27)

.MODEL BD239_106     NPN(Is=2.447p Xti=3 Eg=1.11 Vaf=100 Bf=208.2 Ise=70.69p
+ Ne=1.565 Ikf=.9743 Nk=.6134 Xtb=1.5 Br=12.59 Isc=11.68n
+ Nc=1.835 Ikr=3.86 Rc=.4685 Cjc=142p Mjc=.4353 Vjc=.75 Fc=.5
+ Cje=188.5p Mje=.4878 Vje=.75 Tr=194.2n Tf=19.85n Itf=164.1
+ Xtf=5.945 Vtf=10 Rb=.1 T_ABS=106)

.model Q2N2222_27   NPN(Is=14.34f Xti=3 Eg=1.11 Vaf=74.03 Bf=255.9 Ne=1.307
+ Ise=14.34f Ikf=.2847 Xtb=1.5 Br=6.092 Nc=2 Isc=0 Ikr=0 Rc=1
+ Cjc=7.306p Mjc=.3416 Vjc=.75 Fc=.5 Cje=22.01p Mje=.377 Vje=.75
+ Tr=46.91n Tf=411.1p Itf=.6 Vtf=1.7 Xtf=3 Rb=10 T_ABS=27)
*

.temp=143 ; 27 ; pour les 2 Darlington
* circuit :
VDC1puiss VPLUS 0 50 ; alimentation
VDC2puiss VMOINS 0 -50 ; parfaite

XQ1 VPLUS B1 E1 Q2N6059 ; C B E darlington NPN
R1 E1 S 0.1
R2 S E2 0.15
XQ2 VMOINS B2 E2 Q2N6052 ; C B E darlington PNP
RL S 0 8 ; haut parleur

R3 VPLUS B1 2.2k
R5 B1 B3 2.2k
R6 B3 B2 2.2k
Q4 B2 B4 E4 Q2N2222_27 ; ne chauffe pas
R7 VPLUS B4 47k
R4 E4 VMOINS 100
R8 B4 VMOINS 1.5k
.param A=1.5
VE in 0 DC=0 sin (0 {A} 1k 1m 0 0) AC 1
C1 in B4 10u ; arrivée du signal

Q3 B1 B3 B2 BD239_106 ; 27 ; C B E

.OP
.TRAN 10u 1.01 0 10u ;
*.STEP param A list 0 0.2 0.4 0.6
.probe
.end
```

Netliste, prête à simuler

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphasé sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

